

Tykillä ampuminen 1

Heittoliikkeeseen kuuluu keskeisinä tekijöinä maan vetovoima sekä ilmanvastuksen aiheuttama nopeudelle vastakkaissuuntainen voima. Näitä voimia sovelletaan myös mallinnettaessa tykin ammuksen liikerataa.

Yksinkertaisimmassa mallissa jätetään ilmanvastus huomiotta. Tällöin ammuksen vaikuttaa lähtöhetken jälkeen vain maan vetovoima $\vec{G} = m\vec{g}$, missä $\vec{g}$ on maan vetovoiman aiheuttama

kiihtyvyys ja m ammuksen massa. Kun kiihtyvyys lausutaan derivaatan avulla $\vec{a} = \frac{\partial^2 \vec{r}}{\partial t^2}$, saa yleinen liikeyhtälö $\vec{F} = m\vec{a}$ tällöin muodon

$$m \frac{\partial^2 \vec{r}}{\partial t^2} = m\vec{g},$$

mistä massa m voidaan supistaa pois.

Tarkastellaan seuraavaa ongelmaa: Mikä on oikea ampumiskulma, kun 25 metrin korkeudella sijaitsevalta tykkilavetiltä ammutaan 10 kg kranaatti nopeudella 500 m/s kohti meren pinnalla olevaa maalia, jonka etäisyys on tasan 10 km?

Jaetaan yllä esitetty toisen kertaluvun differentiaaliyhtälö kahteen eri osayhtälöön, z-suuntaan ja x-suuntaan, ja ratkaistaan yhtälöt. Laskujen aluksi on syytä hävittää mahdollisista aiemmista laskuista jääneet muuttujat.

```
> restart;
```

Määritellään differentiaaliyhtälöryhmä ja sen tuntemattomat muuttujat.

```
> ryhmä := diff(x(t), t$2)=0, diff(z(t), t$2)=-g;
```

$$ryhmä := \frac{d^2}{dt^2} x(t) = 0, \frac{d^2}{dt^2} z(t) = -g$$

```
> muuttujat := x(t), z(t);
```

$$muuttujat := x(t), z(t)$$

Määritellään vakiot ja alkuehto. Ammuksen lähtökulma olkoon θ . Sisällytetään ehtoon sekä ammuksen alkunopeus jaettuna nopeuskomponentteihin että lähtökorkeus.

```
> g:= 9.81: m:= 10: v0:= 500:
```

```
> alkuehto:= x(0)=0, z(0)=25, D(x)(0)=v0*cos(theta),  
D(z)(0)=v0*sin(theta);
```

$$alkuehto := x(0) = 0, z(0) = 25, D(x)(0) = 500 \cos(\theta), D(z)(0) = 500 \sin(\theta)$$

Ratkaistaan alkuarvoprobleema:

```
> dyrtk:= dsolve({ryhma, alkuehto}, {muuttujat});
```

$$\text{dyrtk} := \left\{ z(t) = -\frac{981 t^2}{200} + 500 \sin(\theta) t + 25, x(t) = 500 \cos(\theta) t \right\}$$

Ammuksen rata on paraabeli, jonka parametrimuotoinen esitys on

```
> lentorata:= subs(dyrtk, [x(t), z(t)]);
```

$$\text{lentorata} := \left[500 \cos(\theta) t, -\frac{981 t^2}{200} + 500 \sin(\theta) t + 25 \right]$$

Ammuksen kokonaislentoaika ja lähtökulma saadaan ratkaisemalla yhtälöryhmä:

```
> sys:= zip((x, y)->x=y, lentorata, [10000, 0]);
```

$$\text{sys} := \left[500 \cos(\theta) t = 10000, -\frac{981 t^2}{200} + 500 \sin(\theta) t + 25 = 0 \right]$$

```
> rtk:= solve({sys[]}, {t, theta});
```

$$\begin{aligned} \text{rtk} := & \left\{ \theta = \arctan\left(\frac{962361}{640004000} \text{RootOf}(962361 _Z^4 - 4003924 _Z^2 + 640004, \text{label} = _L3)\right)^3 \right. \\ & + \frac{154959019}{32002000} \text{RootOf}(962361 _Z^4 - 4003924 _Z^2 + 640004, \text{label} = _L3), \\ & - \frac{962361}{1600010} \text{RootOf}(962361 _Z^4 - 4003924 _Z^2 + 640004, \text{label} = _L3)^3 \\ & \left. + \frac{2001962}{800005} \text{RootOf}(962361 _Z^4 - 4003924 _Z^2 + 640004, \text{label} = _L3) \right\}, \\ & t = 50 \text{RootOf}(962361 _Z^4 - 4003924 _Z^2 + 640004, \text{label} = _L3) \end{aligned}$$

RootOf(...) edustaa jotakin yhtälön juurta. Juuret saadaan ratkaistua *allvalues*-komennolla.

```
> juuret:= [evalf(allvalues(rtk))];
```

$$\text{juuret} := \left[\{ t = -99.92514551, \theta = -1.772307162 \}, \{ t = 99.92514551, \theta = 1.369285492 \}, \right. \\ \left. \{ t = -20.40269588, \theta = -2.942581815 \}, \{ t = 20.40269588, \theta = 0.1990108389 \} \right]$$

Vain positiiviset ratkaisut kelpaavat. Mahdollisuuksia on siis kaksi. Näitä vastaavat lentoradat ovat

```
> radat:= select(x->subs(x, t)>0, juuret);
```

$$\text{radat} := \left[\{ t = 99.92514551, \theta = 1.369285492 \}, \{ t = 20.40269588, \theta = 0.1990108389 \} \right]$$

```
> rata1:= subs(theta=subs(radat[1], theta), lentorata);
```

$$\text{rata1} := \left[500 \cos(1.369285492) t, -\frac{981 t^2}{200} + 500 \sin(1.369285492) t + 25 \right]$$

```
> rata2:= subs(theta=subs(radat[2], theta), lentorata);
```

$$rata2 := \left[500 \cos(0.1990108389) t, -\frac{981 t^2}{200} + 500 \sin(0.1990108389) t + 25 \right]$$

Sekunneissa ilmoitetut kokonaislentoajat ovat

```
> ajat:= map(x->subs(x, t), radat);
```

```
ajat := [99.92514551, 20.40269588]
```

Lähtökulmat ovat radiaaneissa. Asteissa ilmoitettuina ne ovat


```
> map(x->subs(x, theta), radat):
```

```
evalf(map(x->convert(x, degrees), %));
```


```
[78.45427963 degrees, 11.40248114 degrees]
```

Radoista voidaan piirtää kuvaajat ja yhdistää nämä samaan kuvaan:


```
> kuva1:= plot([rata1[], t=0..ajat[1]], color=green):  
kuva1;
```



```
> kuva2:= plot([rata2[], t=0..ajat[2]], color=red):  
kuva2;
```



```
> with(plots):  
Warning, the name changecoords has been redefined  
  
> display(kuva1, kuva2);
```



>

Tehtäviä

Määritä ampumaetäisyys lähtökulman funktiona. Piirrä funktion kuvaaja. Millä kulmalla saadaan suurin ampumaetäisyys? Mikä tämä etäisyys on? Kauanko ammuksen lento tällöin kestää?

Miten radan lakikorkeus riippuu lähtökulmasta? Entä ampumaetäisyydestä? Piirrä kuvaajat!

Linkit

tykillä ampuminen ilmanvastus huomioiden (tykki2.mws)

JP & SKK & MS 12.07.2001